

Hi Tewodoros

1. I recommend that you start by reading the following page:  
<http://www.dhsprogram.com/data/Merging-Datasets.cfm>

Then sort the datasets you are using according to the variables that will be used for merging:  
Household (HR): sort cases by HV001 HV002.  
Couple (CR): sort cases by V001 V002 V003.  
HIV test results: sort cases by HIVCLUST HIVNUMB HIVLINE.

Let's next take the example of merging the household (HR) data with the couple's data. You need the following step:

```
match files
  /file=*
  /table='xxhr60fl_sorted.sav'
  /rename (hv001,hv002=v001,v002)
  /by v001 v002.
execute.
```

where xxhr60fl\_sorted.sav is the sorted version of the HR dataset.

Then, let's take the example of merging the HIV test results with the couple's data. you have to do this in two steps - first the women and then the men.

For the women, we will create renamed variables with the women's test results.

```
match files
  /file=*
  /table='xxar60fl_sorted.sav'
  /rename (hivclust,hivnumb,hivline,hiv01,hiv02,hiv03,hiv05=v001,v002,
v003,whiv01,whiv02,whiv03,whiv05)
  /by v001 v002 v003.
execute.
```

where xxar60fl\_sorted.sav is the sorted version of the HIV test results dataset.

Now for the men, you first need to re-sort the couples data by the men's line number within the cluster and household:  
sort cases by v001 v002 v034.

And finally merge the HIV test results for the men, creating separate renamed variables for the men's test results.

```
match files
  /file=*
  /table='xxar60fl_sorted.sav'
  /rename (hivclust,hivnumb,hivline,hiv01,hiv02,hiv03,hiv05=v001,v002,
```

```
v034,mhiv01,mhiv02,mhiv03,mhiv05)
  /by v001 v002 v034.
execute.
```

To test the result, you need to first weight the data. For the couple's data we generally use the men's weight (mv005), however, when tabulating the HIV test results for couples, we use the HIV weight from the men's HIV test result (in this example, that is mhiv05, renamed from hiv05).  
compute wgt = mhiv05/1000000.  
weight by wgt.

Finally you can crosstabulate whiv03 with mhiv03 and compare the results to the table on couple's HIV status in the DHS reports.

I did this for the Zimbabwe DHS 2010-11 survey, comparing to table 14.12 in the report. See the attached example program.

2. If you follow the example above, then you will be able to select the cases where (whiv03 = 0 and mhiv03 = 1) or (whiv03 = 1 and mhiv03 = 0), i.e. where women's and men's HIV test results differed, but both were tested.

### File Attachments

1) [Merging example syntax.sps](#), downloaded 829 times

---